
Linear perturbation of the metric 

tensor

Preliminaries

Load tensor package

In[2]:= << xAct`xPert`

------------------------------------------------------------

Package xAct`xPerm` version 1.2.3, {2015 , 8, 23 }

CopyRight (C) 2003 -2020 , Jose M. Martin -Garcia , under the General Public License .

Connecting to external linux executable ...

Connection established .

------------------------------------------------------------

Package xAct`xTensor` version 1.1.4, {2020 , 2, 16 }

CopyRight (C) 2002 -2020 , Jose M. Martin -Garcia , under the General Public License .

------------------------------------------------------------

Package xAct`xPert` version 1.0.6, {2018 , 2, 28 }

CopyRight (C) 2005 -2018 , David Brizuela , Jose M. Martin -Garcia

and Guillermo A. Mena Marugan , under the General Public License .

------------------------------------------------------------

These packages come with ABSOLUTELY NO WARRANTY ; for details type

Disclaimer []. This is free software , and you are welcome to redistribute

it under certain conditions . See the General Public License for details .

------------------------------------------------------------

** Variable $PrePrint assigned value ScreenDollarIndices

** Variable $CovDFormat changed from Prefix to Postfix

** Option AllowUpperDerivatives of ContractMetric changed from False to True

** Option MetricOn of MakeRule changed from None to All

** Option ContractMetrics of MakeRule changed from False to True

Nicer printing

In[3]:= $PrePrint = ScreenDollarIndices ;



In[4]:= $CovDFormat = "Prefix";

Object definitions

Spacetime manifold

The  spacetime  manifold  M,  on  which  tensors  will  be  defined.  Some  Greek  letters  are  defined  as 

tangent  space  indices.

In[5]:= DefManifold [M, 4, {α, β, γ, μ, ν, ρ, σ, τ, κ, ξ, ζ }]

** DefManifold : Defining manifold M.

** DefVBundle : Defining vbundle TangentM .

Metric

The  metric  g of signature  (-,+,+,+).  The  Levi-Civita  derivative  of a tensor  Aμ  will  be  written  as ▽μ Aν  in 

prefix  notation  or Aν ;μ  in postfix  notation.

In[6]:= DefMetric [{3, 1, 0}, Met[-μ, -ν], CD, {";", "▽"}, PrintAs → "g"]

** DefTensor : Defining symmetric metric tensor Met [- μ, - ν ].
** DefTensor : Defining antisymmetric tensor epsilonMet [- α, - β, - γ, - ζ ].
** DefTensor : Defining tetrametric TetraMet [- α, - β, - γ, - ζ ].
** DefTensor : Defining tetrametric TetraMet† [- α, - β, - γ, - ζ ].
** DefCovD : Defining covariant derivative CD [- μ ].
** DefTensor : Defining vanishing torsion tensor TorsionCD [α, - β, - γ ].
** DefTensor : Defining symmetric Christoffel tensor ChristoffelCD [α, - β, - γ ].
** DefTensor : Defining Riemann tensor RiemannCD [- α, - β, - γ, - ζ ].
** DefTensor : Defining symmetric Ricci tensor RicciCD [- α, - β ].
** DefCovD : Contractions of Riemann automatically replaced by Ricci .

** DefTensor : Defining Ricci scalar RicciScalarCD [].

** DefCovD : Contractions of Ricci automatically replaced by RicciScalar .

** DefTensor : Defining symmetric Einstein tensor EinsteinCD [- α, - β ].
** DefTensor : Defining Weyl tensor WeylCD [- α, - β, - γ, - ζ ].
** DefTensor : Defining symmetric TFRicci tensor TFRicciCD [- α, - β ].
** DefTensor : Defining Kretschmann scalar KretschmannCD [].

** DefCovD : Computing RiemannToWeylRules for dim 4

** DefCovD : Computing RicciToTFRicci for dim 4

** DefCovD : Computing RicciToEinsteinRules for dim 4

** DefTensor : Defining weight +2 density DetMet []. Determinant .

Perturbations  of the  metric  are  written  as δg.

In[7]:= DefMetricPerturbation [Met, δMet, ϵ, PrintAs → "δg"]

** DefParameter : Defining parameter ϵ.
** DefTensor : Defining tensor δMet [LI [order ], - α, - β ].

Minkowski metric

We  also  define  the  Minkowski  metric  as a flat  metric:

In[  ]:= DefMetric [{3, 1, 0}, Mink [-μ, -ν], PD, {",", "∂"}, FlatMetric → True, PrintAs → "η"]

DefMetric : There are already metrics {Met } in vbundle M.

** DefTensor : Defining symmetric metric tensor Mink [- μ, - ν ].
** DefTensor : Defining inverse metric tensor InvMink [μ, ν ]. Metric is frozen !

** DefTensor : Defining antisymmetric tensor epsilonMink [- α, - β, - γ, - ζ ].
** DefTensor : Defining tetrametric TetraMink [- α, - β, - γ, - ζ ].
** DefTensor : Defining tetrametric TetraMink† [- α, - β, - γ, - ζ ].
** DefMetric : Associating fiducial flat derivative PD to metric .

** DefTensor : Defining weight +2 density DetMink []. Determinant .

Taylor series expansion

Definition

Before  studying  perturbations  of the  metric  tensor  and  other  tensor  derived  from  the  metric,  it is 

helpful  to briefly  recall  the  Taylor  expansion  of a function  f around  some  value  x:

In[9]:= Series [f[x + δx], {δx, 0, 3}]

Out[  ]= f[x] + f′
[x] δx +

1

2
f′′

[x] δx2 +
1

6
f3

[x] δx3 + O[δx]4

In the  following,  we  will  consider  only  the  series  up  to the  first  term,  which  is linear  in the  perturba -

tion  δx of the  argument:

In[  ]:= Series [f[x + δx], {δx, 0, 1}]

Out[  ]= f[x] + f′
[x] δx + O[δx]2

This  is the  coefficient  of this  term

In[  ]:= SeriesCoefficient [f[x + δx], {δx, 0, 1}]

Out[  ]= f′
[x]

Product rule

Since  the  linear  perturbation  is given  by the  derivative,  the  product  rule  holds:

In[12]:= SeriesCoefficient [f[x + δx] × g[x + δx], {δx, 0, 1}]

Out[  ]= g[x] f′
[x] + f[x] g′

[x]

Inverse rule

Similarly  to the  product  rule,  we  have  the  rule

In[13]:= SeriesCoefficient [1 / f[x + δx], {δx, 0, 1}]

Out[  ]= -
f′[x]

f[x]2

Notational  conventions

Perturbation of the metric tensor

The  metric  is expanded  in a formal  Taylor  series  expansion,  where  the  n’th  perturbation  order  

carries  a factor  
ϵn
n!

 and  is written  as δgμν
n . Here  we  will  be  interested  only  in the  linear  perturbations.

In[14]:= Perturbed [Met[-μ, -ν], 3]

Out[  ]= g μ ν + ϵ δg1 μ ν +
1

2
ϵ2 δg2 μ ν +

1

6
ϵ3 δg3 μ ν

This  gives  us a particular  perturbation  order.

In[15]:= Perturbation [Met[-μ, -ν], 2]

Out[  ]= δg2 μ ν

Perturbation of other tensors

For  other  tensors,  the  n’th  order  perturbation  is written  as △n[X ]:
In[16]:= Perturbation [RicciCD [-μ, -ν], 2]

Out[  ]= △2R[▽] μ ν 

The  formal  series  expansion  has  the  same  form:

In[17]:= Perturbed [RicciCD [-μ, -ν], 3]

Out[  ]= ϵ △R[▽] μ ν  +
1

2
ϵ2 △2R[▽] μ ν  +

1

6
ϵ3 △3R[▽] μ ν  + R[▽] μ ν

Perturbations

Inverse metric

We  start  by  calculating  the  perturbation  △g μ ν  of  inverse  gμν  of  the  metric  tensor.  This  can  be 

obtained  from  two  steps.  First,  note  the  relation  between  the  metric  and  its  inverse,  whose  contrac -

tion  yields  a Kronecker  symbol:

In[18]:= Inactive [Met[μ, ρ]] × Inactive [Met[-ρ, -ν]]

Activate [%]

Out[  ]= Inactive  g μ ρ  × Inactive  g ρ ν 

Out[  ]= δ μ
ν

Second,  since  the  Kronecker  symbol  does  not  depend  on  the  metric,  its  perturbation  vanishes:

In[20]:= Perturbation [delta [-μ, ν]]

Out[  ]= 0

We  can  use  this  to calculate  the  perturbation  of the  inverse  metric.  Here  we  use  the  product  rule:

In[21]:= Perturbation [Inactive [Met[μ, ρ]] × Inactive [Met[-ρ, -σ ]]]

Activate [%]

Expand [Met[ν, σ ] %]

Solve [% ⩵ 0, Perturbation [Met[μ, ν]]][[1]]

Out[  ]= Inactive  g ρ σ  △Inactive g μ ρ  + Inactive  g μ ρ  △Inactive g ρ σ 

Out[  ]= g ρ σ △g μ ρ  + g μ ρ δg1 ρ σ

Out[  ]= △g μ ν  + g μ ρ g ν σ δg1 ρ σ

Out[  ]= △g μ ν  → - g μ ρ g ν σ δg1 ρ σ 

We  can  also  do  so explicitly  using  the  Taylor  expansion  in the  perturbation  parameter:

In[29]:= Perturbed [Met[μ, ρ], 1] × Perturbed [Met[-ρ, -σ ], 1]

SeriesCoefficient [%, {ϵ, 0, 1}]

Expand [Met[ν, σ ] %]

Solve [% ⩵ 0, Perturbation [Met[μ, ν]]][[1]]

Out[  ]=  g μ ρ
+ ϵ △g μ ρ   g ρ σ + ϵ δg1 ρ σ 

Out[  ]= g ρ σ △g μ ρ  + g μ ρ δg1 ρ σ

Out[  ]= △g μ ν  + g μ ρ g ν σ δg1 ρ σ

Out[  ]= △g μ ν  → - g μ ρ g ν σ δg1 ρ σ 

In other  words,  the  perturbation  of the  inverse  metric  is given  by taking  the  negative  of the  perturba -

tion  of the  metric,  and  then  raising  its  indices.  This  confirms  our  result:

In[  ]:= Perturbation [Met[μ, ν]]

ExpandPerturbation [%]

SeparateMetric [][%]

Out[  ]= △g μ ν 

Out[  ]= - δg1 μ ν

Out[  ]= - g μ α g ν β δg1α β

Christoffel symbols

We  then  come  to the  perturbation  of the  Christoffel  symbols.  Their  definition  is well-known:

In[  ]:= ChristoffelToGradMetric [ChristoffelCD [μ, -ρ, -ν]]

Out[  ]=

1

2
g μ α - ∂αg ρ ν + ∂νg ρ α + ∂ρg ν α 

Since  we  have  already  calculated  the  perturbation  of the  inverse  metric,  we  can  use  this  formula.  

The  perturbation  of the  derivative  of the  metric  is just  the  derivative  of the  perturbation  (partial  

derivative  and  perturbing  commute):

In[  ]:= Inactive [Perturbation [PD[-ρ][Met[-μ, -ν]]]]

Activate [%]

Out[  ]= Inactive △∂ρg μ ν 

Out[  ]= ∂ρδg1 μ ν

We  then  use  this  rule  to calculate  the  perturbation  of the  Christoffel  symbols.  It is helpful  to realize  

that  the  partial  derivatives  which  appear  can  be rewritten  in terms  of covariant  derivatives,  so that  

in the  end  all  Christoffel  symbols  cancel:

In[39]:= ChristoffelToGradMetric [ChristoffelCD [μ, -ρ, -ν]];

Perturbation [%];

ExpandPerturbation [%]

ChangeCovD [%, PD, CD];

Expand [%];

ContractMetric [%]

ToCanonical [%];

Simplify [%]

SeparateMetric [][%]

Out[  ]=

1

2
- δg1 μ α - ∂αg ρ ν + ∂νg ρ α + ∂ρg ν α  + g μ α - ∂αδg1 ρ ν + ∂νδg1 ρ α + ∂ρδg1 ν α 

Out[  ]=

1

2
Γ[▽]α ν ρ δg1 μ

α +
1

2
Γ[▽]α ρ ν δg1 μ

α -
1

2
Γ[▽]α μ

ρ δg1α ν -

1

2
Γ[▽]α ν ρ δg1 μ

α -
1

2
Γ[▽]α ρ ν δg1 μ

α +
1

2
Γ[▽] ν α ρ δg1 μ α

-
1

2
Γ[▽] ν ρ α δg1 μ α

+

1

2
Γ[▽] ρ α ν δg1 μ α

-
1

2
Γ[▽] ρ ν α δg1 μ α

+
1

2
Γ[▽]α μ

ρ δg1 ν α -
1

2
Γ[▽]α μ

ν δg1 ρ α +

1

2
Γ[▽]α μ

ν δg1 ρ α -
1

2
▽μ δg1 ρ ν  +

1

2
▽ν δg1 μ

ρ  +
1

2
▽ρ δg1 μ

ν 

Out[  ]=

1

2
- ▽μ δg1 ν ρ  + ▽ν δg1 μ

ρ + ▽ρ δg1 μ
ν 

Out[  ]=

1

2
- g μ α ▽α δg1 ν ρ  + g μ β ▽ν δg1 β ρ  + g μ γ ▽ρ δg1 γ ν 

Hence,  the  perturbation  of the  Christoffel  symbols  is given  as the  covariant  derivative  of the  metric  

perturbation.  This  shows  us an important  property  which  we  will  use  later:  the  perturbation  of the  

Christoffel  symbols  is a tensor.  Again  we  check  our  result:

In[  ]:= Perturbation [ChristoffelCD [μ, -ρ, -ν]]

ExpandPerturbation [%];

SeparateMetric [][%]

Out[  ]= △Γ[▽] μ ρ ν 

Out[  ]=

1

2
- g μ α ▽α δg1 ρ ν  + g μ β ▽ν δg1 β ρ  + g μ γ ▽ρ δg1 γ ν 

Riemann curvature tensor

Next  we  expand  the  Riemann  curvature  tensor.  It is defined  as follows  - note  the  order  of the  

indices,  which  is reversed,  compared  to the  most  common  convention:

In[51]:= RiemannCD [-σ, -ρ, -ν, μ]

ChangeCurvature [%]

Out[  ]= R[▽] μ
σ ρ ν

Out[  ]= Γ[▽]α σ ν Γ[▽] μ ρ α - Γ[▽]α ρ ν Γ[▽] μ σ α + ∂ρΓ[▽] μ σ ν - ∂σ Γ[▽] μ ρ ν

We  can  use  this  formula  to calculate  its  perturbation.  Now  we  use  our  previously  gained  knowledge  

that  the  perturbation  of the  Christoffel  symbol  is a tensor.  We  can  thus  combine  partial  derivatives  

and  Christoffel  symbols  to covariant  derivatives  to get  the  desired  result:

In[53]:= RiemannCD [-σ, -ρ, -ν, μ];

ChangeCurvature [%];

Perturbation [%]

ChangeCovD [%, PD, CD];

ToCanonical [%]

ExpandPerturbation [%];

ToCanonical [%];

Simplify [%]

Out[  ]= - Γ[▽] μ σ α △Γ[▽]α ρ ν  + Γ[▽] μ ρ α △Γ[▽]α σ ν  +
Γ[▽]α σ ν △Γ[▽] μ ρ α  - Γ[▽]α ρ ν △Γ[▽] μ σ α  + ∂ρ△Γ[▽] μ σ ν  - ∂σ△Γ[▽] μ ρ ν 

Out[  ]= ▽ρ △Γ[▽] μ ν σ  - ▽σ △Γ[▽] μ ν ρ 

Out[  ]=

1

2
- ▽ρ ▽μ δg1 ν σ  + ▽ρ ▽ν δg1 μ

σ + ▽ρ ▽σ δg1 μ
ν + ▽σ ▽μ δg1 ν ρ - ▽σ ▽ν δg1 μ

ρ - ▽σ ▽ρ δg1 μ
ν 

We  see  that  the  perturbation  is given  by the  second  covariant  derivative  of the  metric  tensor.  This  

confirms  our  result:

In[  ]:= Perturbation [RiemannCD [-σ, -ρ, -ν, μ]]

ExpandPerturbation [%];

Expand [%];

ToCanonical [%];

Simplify [%]

Out[  ]= △R[▽] μ
σ ρ ν 

Out[  ]=

1

2
- ▽ρ ▽μ δg1 ν σ  + ▽ρ ▽ν δg1 μ

σ + ▽ρ ▽σ δg1 μ
ν + ▽σ ▽μ δg1 ν ρ - ▽σ ▽ν δg1 μ

ρ - ▽σ ▽ρ δg1 μ
ν 

Ricci tensor

We  continue  with  the  Ricci  tensor,  which  is obtained  by contraction  of two  indices  of the  curvature  

tensor:

In[  ]:= Inactive [RiemannCD [-μ, -ρ, -ν, ρ]]

Activate [%]

Out[  ]= Inactive  R[▽] ρ
μ ρ ν 

Out[  ]= R[▽] μ ν

Contracting  two  indices  on  the  perturbation  of the  curvature  tensor  therefore  gives  us the  desired  

result:

In[68]:= Perturbation [RiemannCD [-μ, -ρ, -ν, σ ]]

ExpandPerturbation [%];

Expand [%];

ToCanonical [%];

Simplify [%]

delta [-σ, ρ] %

Expand [%];

ToCanonical [%];

Simplify [%]

Out[  ]= △R[▽] σ
μ ρ ν 

Out[  ]=

1

2
- ▽μ ▽ν δg1 σ

ρ  - ▽μ ▽ρ δg1 σ
ν + ▽μ ▽σ δg1 ν ρ + ▽ρ ▽μ δg1 σ

ν + ▽ρ ▽ν δg1 σ
μ - ▽ρ ▽σ δg1 μ ν 

Out[  ]=

1

2
- ▽μ ▽ν δg1 σ

σ  - ▽μ ▽σ δg1 σ
ν + ▽μ ▽σ δg1 ν σ + ▽σ ▽μ δg1 σ

ν + ▽σ ▽ν δg1 σ
μ - ▽σ ▽σ δg1 μ ν 

Out[  ]=

1

2
- ▽ν ▽μ δg1σ

σ  + ▽σ ▽μ δg1 σ
ν + ▽σ ▽ν δg1 σ

μ - ▽σ ▽σ δg1 μ ν 

Here  two  identical  terms  cancel.  This  confirms  our  result:

In[  ]:= Perturbation [RicciCD [-μ, -ν]]

ExpandPerturbation [%];

Expand [%];

ToCanonical [%];

Simplify [%]

Out[  ]= △R[▽] μ ν 

Out[  ]=

1

2
- ▽α ▽α δg1 μ ν  + ▽α ▽μ δg1 α

ν + ▽α ▽ν δg1 α
μ - ▽ν ▽μ δg1α

α 

Ricci scalar

The  Ricci  scalar  is obtained  by contracting  the  Ricci  tensor  with  the  metric:

In[  ]:= RicciCD [-μ, -ν] × Met[μ, ν]

Out[  ]= R[▽]

Again  we  can  make  use  of our  previous  knowledge  and  the  product  rule:

In[  ]:= Perturbation [Inactive [RicciCD [-μ, -ν]] × Inactive [Met[μ, ν]]];

Activate [%]

ExpandPerturbation [%];

Expand [%];

ToCanonical [%];

Simplify [%]

Out[  ]= g μ ν △R[▽] μ ν  + △g μ ν  R[▽] μ ν

Out[  ]= - R[▽]α μ δg1α μ -
1

2
gα μ ▽μ ▽α δg1 ν

ν - 2 ▽ν ▽μ δg1 ν
α  + ▽ν ▽ν δg1α μ 

This  can  also  be obtained  directly:

In[  ]:= Perturbation [RicciScalarCD []]

ExpandPerturbation [%];

Expand [%];

ContractMetric [%];

ToCanonical [%];

Simplify [%]

Out[  ]= △[R[▽]]
Out[  ]= - R[▽]α β δg1α β + ▽β ▽α δg1α β

- ▽β ▽β δg1α
α

Einstein tensor

Finally,  we  calculate  the  perturbation  of the  Einstein  tensor.  Its  definition  is well-known,  and  so we  

obtains  its  perturbation:

In[  ]:= EinsteinToRicci [EinsteinCD [-μ, -ν]]

Perturbation [%]

ExpandPerturbation [%];

Expand [%];

ContractMetric [%];

ToCanonical [%];

Simplify [%]

Out[  ]= R[▽] μ ν -
1

2
g ν μ R[▽]

Out[  ]= △R[▽] μ ν  +
1

2
- g ν μ △[R[▽]] - R[▽] δg1 ν μ 

Out[  ]=

1

2
-R[▽] δg1 μ ν - ▽α ▽α δg1 μ ν + ▽α ▽μ δg1 α

ν +

▽α ▽ν δg1 α
μ + g μ ν  R[▽]α β δg1α β - ▽β ▽α δg1α β

+ ▽β ▽β δg1α
α  - ▽ν ▽μ δg1α

α 

We  see  that  it is significantly  more  complicated  than  the  perturbation  of the  Ricci  tensor;  hence,  

whenever  we  consider  the  perturbation  of the  vacuum  Einstein  equations,  it is much  simpler  to 

consider  their  trace-reversed  form  Rμν = 0 instead  of Gμν = 0. We  finally  check:

In[  ]:= Perturbation [EinsteinCD [-μ, -ν]]

ExpandPerturbation [%];

Expand [%];

ContractMetric [%];

ToCanonical [%];

Simplify [%]

Out[  ]= △G[▽] μ ν 

Out[  ]=

1

2
-R[▽] δg1 μ ν - ▽α ▽α δg1 μ ν + ▽α ▽μ δg1 α

ν +

▽α ▽ν δg1 α
μ + g μ ν  R[▽]α β δg1α β - ▽β ▽α δg1α β

+ ▽β ▽β δg1α
α  - ▽ν ▽μ δg1α

α 

Perturbations  around the Minkowski  metric

Collecting perturbations

Recall  from  the  previous  section  the  objects  we  studied  and  their  perturbations:

In[  ]:= tens = {Met[-μ, -ν], Met[μ, ν], ChristoffelCD [μ, -ρ, -ν],

RiemannCD [-σ, -ρ, -ν, μ], RicciCD [-μ, -ν], RicciScalarCD [], EinsteinCD [-μ, -ν]}

Perturbation /@ %;

ExpandPerturbation /@ %;

Expand /@ %;

ToCanonical /@ %;

pert = Simplify /@ %;

Out[  ]=  g μ ν , g μ ν , Γ[▽] μ ρ ν , R[▽] μ
σ ρ ν , R[▽] μ ν , R[▽], G[▽] μ ν 

Setting background to Minkowski

In the  following,  we  will  set  the  metric  to the  Minkowski  metric,  and  likewise  for  the  inverse:

In[  ]:= gtoη =

Join [MakeRule [{Met[-μ, -ν], Mink [-μ, -ν]}, MetricOn → None, ContractMetrics → False ],

MakeRule [{Met[μ, ν], InvMink [μ, ν]}, MetricOn → None, ContractMetrics → False ]]

Out[  ]= HoldPattern  g μ ν  ⧴ Module {}, η μ ν , HoldPattern  g μ ν  ⧴ Module {}, iη μ ν 

To translate  the  perturbations  from  a general  background  to a Minkowski  background,  we  perform

the  following  steps:

 Bring  all  indices  to their  natural  position,  insert  metrics  or inverse  metrics  as needed.

 Change  all  curvature  terms  to derivatives  and  products  of Christoffel  symbols.

 Change  all  covariant  derivatives  to partial  derivatives  and  Christoffel  symbols.

 Change  all  Christoffel  symbols  to derivatives  of the  metric.

 Replace  the  metric  and  its  inverse  by the  Minkowski  metric.

In[  ]:= pert;

SeparateMetric [] /@ %;

ChangeCurvature /@ %;

ChangeCovD [#, CD, PD] & /@ %;

ChristoffelToGradMetric /@ %;

% //. gtoη;

Expand /@ %;

ContractMetric /@ %;

ToCanonical [#, UseMetricOnVBundle → None ] & /@ %;

flat = Simplify /@ %;

Comparison of results

The  following  table  compares  the  perturbations  around  a general  background  and  around  the  

Minkowski  metric:

In[  ]:= Grid [Prepend [ScreenDollarIndices [Transpose [{tens, pert, flat}]],

{"Object", "General background metric", "Minkowski background "}], Frame → All]

Out[  ]=

Object General background metric Minkowski background

g μ ν δg1 μ ν δg1 μ ν
g μ ν

- δg1 μ ν
- iη μ α

iη ν β δg1α β

Γ[▽] μ ρ ν
1

2
- ▽μ δg1 ν ρ  + ▽ν δg1 μ

ρ + ▽ρ δg1 μ
ν  1

2
iη μ α - ∂αδg1 ν ρ + ∂νδg1 ρ α + ∂ρδg1 ν α 

R[▽] μ
σ ρ ν

1

2
- ▽ρ ▽μ δg1 ν σ  + ▽ρ ▽ν δg1 μ

σ +

▽ρ ▽σ δg1 μ
ν + ▽σ ▽μ δg1 ν ρ -

▽σ ▽ν δg1 μ
ρ - ▽σ ▽ρ δg1 μ

ν 

1

2
iη μ α - ∂α∂ρδg1 ν σ +

∂α∂σ δg1 ν ρ + ∂ρ∂νδg1σ α - ∂σ∂νδg1 ρ α 

R[▽] μ ν
1

2
- ▽α ▽α δg1 μ ν  + ▽α ▽μ δg1 α

ν +

▽α ▽ν δg1 α
μ - ▽ν ▽μ δg1α

α 

1

2
iη α β - ∂β∂αδg1 μ ν +

∂β∂μδg1 ν α + ∂β∂νδg1 μ α - ∂ν∂μδg1α β 
R[▽] - R[▽]α β δg1α β -

1

2
gα β ▽β ▽α δg1 γ

γ - 2 ▽γ ▽β δg1 γ
α  +

▽γ ▽γ δg1α β 

iη α β
iη γ ζ ∂ζ ∂βδg1α γ - ∂ζ ∂γδg1α β 

G[▽] μ ν
1

4
 g μ ν 2 R[▽]α β δg1α β + gα β

▽β ▽α δg1 γ
γ - 2 ▽γ ▽β δg1 γ

α  +
▽γ ▽γ δg1α β  -

2 R[▽] δg1 μ ν + ▽α ▽α δg1 μ ν -

▽α ▽μ δg1 α
ν -

▽α ▽ν δg1 α
μ + ▽ν ▽μ δg1α

α 

1

2
iη α β - ∂β∂αδg1 μ ν + ∂β∂μδg1 ν α +

∂β∂νδg1 μ α - iη γ ζ η μ ν ∂ζ ∂βδg1α γ +

iη γ ζ η μ ν ∂ζ ∂γδg1α β - ∂ν∂μδg1α β 

